

COMP4075: Lecture 11

Monad Transformers

Henrik Nilsson

University of Nottingham, UK

COMP4075: Lecture 11 – p.1/31

Monad Transformers (2)

However:

- Not always obvious how: e.g., should the combination of state and error have been

$$\text{newtype } SE\ s\ a = SE\ (s \rightarrow (Maybe\ a,\ s))$$

- Duplication of effort: similar patterns related to specific effects are going to be repeated over and over in the various combinations.

COMP4075: Lecture 11 – p.3/31

Monad Transformers (1)

What if we need to support more than one type of effect?

For example: State and Error/Partiality?

We could implement a suitable monad from scratch:

$$\text{newtype } SE\ s\ a = SE\ (s \rightarrow Maybe\ (a,\ s))$$

COMP4075: Lecture 11 – p.2/31

Monad Transformers (3)

Monad Transformers can help:

- A **monad transformer** transforms a monad by adding support for an additional effect.
- Monad transformer libraries can be developed, each transformer each adding a specific effect (state, error, ...).
- A form of **aspect-oriented programming**.
- MTL is one example of such a library.

Will consider the general idea of monad transformers first; specific libraries discussed later.

COMP4075: Lecture 11 – p.4/31

Monad Transformers in Haskell (1)

- A **monad transformer** maps monads to monads. Represented by a type constructor T of the following kind:

$$T :: (* \rightarrow *) \rightarrow (* \rightarrow *)$$

- Additionally, a monad transformer **adds** computational effects.
- A mapping $lift$ maps a computation in the underlying monad to one in the transformed monad:

$$lift :: M\ a \rightarrow T\ M\ a$$

COMP4075: Lecture 11 – p.5/31

Monad Transformers in Haskell (2)

- These requirements are captured by the following (multi-parameter) type class:

```
class (Monad m, Monad (t m))
  => MonadTransformer t m where
  lift :: m a -> t m a
```

COMP4075: Lecture 11 – p.6/31

Classes for Specific Effects

A monad transformer adds specific effects to **any** monad. Thus the effect-specific operations needs to be overloaded. For example:

```
class Monad m => E m where
```

```
  eFail    :: m a
```

```
  eHandle :: m a -> m a -> m a
```

```
class Monad m => S m s | m -> s where
```

```
  sSet :: s -> m ()
```

```
  sGet :: m s
```

COMP4075: Lecture 11 – p.7/31

The Identity Monad

We are going to construct monads by successive transformations of the identity monad:

```
newtype I a = I a
```

```
unI (I a) = a
```

```
instance Monad I where
```

```
  return a = I a
```

```
  m >>= f = f (unI m)
```

```
runI :: I a -> a
```

```
runI = unI
```

COMP4075: Lecture 11 – p.8/31

The Error Monad Transformer (1)

```
newtype ET m a = ET (m (Maybe a))  
unET (ET m) = m
```

The Error Monad Transformer (2)

Any monad transformed by *ET* is a monad:

```
instance Monad m  $\Rightarrow$  Monad (ET m) where  
  return a = ET (return (Just a))  
  m  $\gg=$  f = ET $ do  
    ma  $\leftarrow$  unET m  
  case ma of  
    Nothing  $\rightarrow$  return Nothing  
    Just a  $\rightarrow$  unET (f a)
```

COMP4075: Lecture 11 – p.9/31

COMP4075: Lecture 11 – p.10/31

The Error Monad Transformer (3)

We need the ability to run transformed monads:

```
runET :: Monad m  $\Rightarrow$  ET m a  $\rightarrow$  m a  
runET etm = do  
  ma  $\leftarrow$  unET etm  
  case ma of  
    Just a  $\rightarrow$  return a  
    Nothing  $\rightarrow$  error "Should not happen"
```

(Note: To simplify use, we discarded information about the effect, but as a result, we get a partial function. Returning *Maybe a* better in general.)

COMP4075: Lecture 11 – p.11/31

The Error Monad Transformer (4)

ET is a monad transformer:

```
instance Monad m  $\Rightarrow$   
  MonadTransformer ET m where  
  lift m = ET (m  $\gg=$   $\lambda$ a  $\rightarrow$  return (Just a))
```

COMP4075: Lecture 11 – p.12/31

The Error Monad Transformer (5)

Any monad transformed by ET is an instance of E :

```
instance Monad m => E (ET m) where
  eFail = ET (return Nothing)
  m1 `eHandle` m2 = ET $ do
    ma <- unET m1
  case ma of
    Nothing -> unET m2
    Just _   -> return ma
```

COMP4075: Lecture 11 – p.13/31

Exercise 1: Running Transf. Monads

Let

```
ex2 = eFail `eHandle` return 1
```

1. Suggest a possible type for $ex2$.
(Assume $1 :: Int$.)
2. Given your type, use the appropriate combination of “run functions” to run $ex2$.

COMP4075: Lecture 11 – p.15/31

The Error Monad Transformer (6)

A state monad transformed by ET is a state monad:

```
instance S m s => S (ET m) s where
  sSet s = lift (sSet s)
  sGet  = lift sGet
```

COMP4075: Lecture 11 – p.14/31

Exercise 1: Solution

```
ex2 :: ET I Int
ex2 = eFail `eHandle` return 1
ex2result :: Int
ex2result = runI (runET ex2)
```

COMP4075: Lecture 11 – p.16/31

The State Monad Transformer (1)

```
newtype ST s m a = ST (s → m (a, s))
unST (ST m) = m
```

Any monad transformed by ST is a monad:

```
instance Monad m ⇒ Monad (ST s m) where
  return a = ST (λs → return (a, s))
  m ≫= f = ST $ λs → do
    (a, s') ← unST m s
    unST (f a) s'
```

COMP4075: Lecture 11 – p.17/31

The State Monad Transformer (2)

We need the ability to run transformed monads:

```
runST :: Monad m ⇒ ST s m a → s → m a
runST stf s0 = do
  (a, _) ← unST stf s0
  return a
```

(We are again discarding information to keep things simple. Returning the final state along with result would be more general.)

COMP4075: Lecture 11 – p.18/31

The State Monad Transformer (3)

ST is a monad transformer:

```
instance Monad m ⇒
  MonadTransformer (ST s) m where
  lift m = ST (λs → m ≫= λa → return (a, s))
```

COMP4075: Lecture 11 – p.19/31

The State Monad Transformer (3)

Any monad transformed by ST is an instance of S :

```
instance Monad m ⇒ S (ST s m) s where
  sSet s = ST (λ_ → return ((), s))
  sGet = ST (λs → return (s, s))
```

COMP4075: Lecture 11 – p.20/31

The State Monad Transformer (4)

An error monad transformed by ST is an error monad:

instance $E\ m \Rightarrow E\ (ST\ s\ m)$ **where**
 $eFail = lift\ eFail$
 $m1\ 'eHandle'\ m2 = ST\ \$\ \lambda s \rightarrow$
 $unST\ m1\ s\ 'eHandle'\ unST\ m2\ s$

COMP4075: Lecture 11 – p.21/31

Exercise 2: Solution

$$\begin{aligned} runI\ (runET\ (runST\ ex3a\ 0)) &= 0 \\ runI\ (runST\ (runET\ ex3b)\ 0) &= 42 \end{aligned}$$

Why? Because:

$$\begin{aligned} ST\ s\ (ET\ I)\ a &\cong s \rightarrow (ET\ I)\ (a, s) \\ &\cong s \rightarrow I\ (Maybe\ (a, s)) \\ &\cong s \rightarrow Maybe\ (a, s) \\ ET\ (ST\ s\ I)\ a &\cong (ST\ s\ I)\ (Maybe\ a) \\ &\cong s \rightarrow I\ (Maybe\ a, s) \\ &\cong s \rightarrow (Maybe\ a, s) \end{aligned}$$

COMP4075: Lecture 11 – p.23/31

Exercise 2: Effect Ordering

Consider the code fragment

$$\begin{aligned} ex3a &:: (ST\ Int\ (ET\ I))\ Int \\ ex3a &= (sSet\ 42 \gg eFail)\ 'eHandle'\ sGet \end{aligned}$$

Note that the exact same code fragment also can be typed as follows:

$$\begin{aligned} ex3b &:: (ET\ (ST\ Int\ I))\ Int \\ ex3b &= (sSet\ 42 \gg eFail)\ 'eHandle'\ sGet \end{aligned}$$

What is

$$\begin{aligned} runI\ (runET\ (runST\ ex3a\ 0)) \\ runI\ (runST\ (runET\ ex3b)\ 0) \end{aligned}$$

COMP4075: Lecture 11 – p.22/31

MTL: Monad Transformer Library

Provides a number of standard monads, associated transformers, and all possible liftings in the style we have seen; e.g.:

- State (*Control.Monad.State*, lazy and strict)
- Exceptions (*Control.Monad.Except*)
- Lists (*Control.Monad.List*)
- Reader (*Control.Monad.Reader*)
- Writer (*Control.Monad.Writer*)
- Continuations (*Control.Monad.Cont*)

COMP4075: Lecture 11 – p.24/31

MTL: State

```
class Monad m => MonadState s m | m -> s where
  get  :: m s
  put  :: s -> m ()
  state :: (s -> (a, s)) -> m a
```

Transformer: `newtype StateT s (m :: * -> *) a`

Run functions:

```
runState :: State s a -> s -> (a, s)
evalState :: State s a -> s -> a
execState :: State s a -> s -> s
```

COMP4075: Lecture 11 - p.25/31

MTL: Exception

```
class Monad m =>
  MonadError e m | m -> e where
  throwError :: e -> m a
  catchError :: m a -> (e -> m a) -> m a
```

Transformer: `newtype ExceptT e (m :: * -> *) a`

Run function:

```
runExcept :: Except e a -> Either e a
```

COMP4075: Lecture 11 - p.26/31

MTL: Reader

```
class Monad m =>
  MonadReader r m | m -> r where
  ask  :: m r
  local :: (r -> r) -> m a -> m a
  reader :: (r -> a) -> m a
```

Transformer: `ReaderT`

Run function:

```
runReader :: Reader r a -> r -> a
```

COMP4075: Lecture 11 - p.27/31

MTL: Writer

```
class (Monoid w, Monad m) =>
  MonadWriter w m | m -> w where
  writer :: (a, w) -> m a
  tell   :: w -> m ()
  listen :: m a -> m (a, w)
  pass   :: m (a, w -> w) -> m a
```

Transformer: `newtype WriterT w (m :: * -> *) a`

Run function:

```
runWriter :: Writer w a -> (a, w)
```

COMP4075: Lecture 11 - p.28/31

Problems with Monad Transformers

- With one transformer for each possible effect we get a quadratic number of combinations; each has to be instantiated explicitly.
- Jaskelioff (2008,2009) has proposed a possible, more extensible alternative:
 - Traditional approach: unsystematic lifting on case-by-case basis.
 - Jaskelioff: systematic lifting based on theoretical principles where each operation is paired with a type of its implementation allowing implementations to be transformed generically.

COMP4075: Lecture 11 – p.29/31

Reading (1)

- Nick Benton, John Hughes, Eugenio Moggi. Monads and Effects. In *International Summer School on Applied Semantics 2000*, Gaminha, Portugal, 2000.
- Sheng Liang, Paul Hudak, Mark Jones. Monad Transformers and Modular Interpreters. In *Proceedings of the 22nd ACM Symposium on Principles of Programming Languages (POPL'95)*, January 1995, San Francisco, California

COMP4075: Lecture 11 – p.30/31

Reading (2)

- Mauro Jaskelioff. Monatron: An Extensible Monad Transformer Library. In *Implementation of Functional Languages (IFL'08)*, 2008.
- Mauro Jaskelioff. Modular Monad Transformers. In *European Symposium on Programming (ESOP,09)*, 2009.

COMP4075: Lecture 11 – p.31/31